Low-Discrepancy Sequences

Van der Corput and Halton Sequences

One of the classical low-discrepancy sequences is the van der Corput sequence in base b, where b is any integer greater than one. To find the nth term, x_n, of the sequence, we first write the unique digit expansion of n in base b: $n = \sum_{j=0}^{\infty} a_j(n)\, b^j$. This is the unique expansion that has only finitely many non-zero coefficients $a_j(n)$. The next step is to evaluate $\phi_b(n)$, the radical inverse function in base b, which is defined as : $\phi_b(n) = \sum_{j=0}^{\infty} a_j(n)\, b^{-j-1}$. The van der Corput sequence in base b is then given by $x_n = \phi_b(n)$, for all n ≥ 0.

Let us illustrate in detail how we can use *Mathematica* to compute $\phi_2(6)$, a particular term of the van der Corput sequence in base 2. First we need to write 6 in its binary expansion. This can be done using the built-in *Mathematica* function, **IntegerDigits**. Here is the digit expansion of 6 in base 2: **IntegerDigits[6,2] = {1, 1, 0}**. The next step is to calculate the sum $a_2(n)\, 2^{-3} + a_1(n)\, 2^{-2} + a_0(n)\, 2^{-1}$, where $a_0 = 0,\ a_1 = 1, a_2 = 1$. This is done by applying the *Mathematica* function Dot and using the listability of the exponent "^". Here is the general formula for any number n and base b.

```
In[1]:= vanderCorput[0, b_] = 0;

In[2]:= vanderCorput[n_, b_] := IntegerDigits[n, b].(b^Range[-Floor[Log[b, n] + 1], -1])

        Attributes[vanderCorput] = Listable;
```

These are the first 10 van der Corput numbers in base 2, starting from n=1.

```
Table[vanderCorput[n, 2], {n, 10}]
```

$$\left\{\frac{1}{2}, \frac{1}{4}, \frac{3}{4}, \frac{1}{8}, \frac{5}{8}, \frac{3}{8}, \frac{7}{8}, \frac{1}{16}, \frac{9}{16}, \frac{5}{16}\right\}$$

Van der Corput numbers in pairwise relatively prime bases can be used to construct low-discrepancy sequences in higher dimensions. The s-dimensional Halton sequence in bases b_1,...,b_s is defined by, $x_n = (\phi_{b_1}(n), \phi_{b_2}(n),\ \ldots, \phi_{b_s}(n))$, where b_1,...,b_s are pairwise relatively prime numbers. Our choice for b_i is the ith prime number, which can be generated in *Mathematica* using the **Prime** function. The following code generates s-dimensional Halton vectors in bases p_1,...,p_s, where p_i is the ith prime number.

```
In[4]:= halton[n_Integer, s_Integer] :=
            vanderCorput[n, Prime[Range[s]]];
```

Here are the first six three-dimensional Halton vectors:

```
Table[halton[n, 3], {n, 6}]
```

$$\left\{\left\{\frac{1}{2}, \frac{1}{3}, \frac{1}{5}\right\}, \left\{\frac{1}{4}, \frac{2}{3}, \frac{2}{5}\right\}, \left\{\frac{3}{4}, \frac{1}{9}, \frac{3}{5}\right\}, \left\{\frac{1}{8}, \frac{4}{9}, \frac{4}{5}\right\}, \left\{\frac{5}{8}, \frac{7}{9}, \frac{1}{25}\right\}, \left\{\frac{3}{8}, \frac{2}{9}, \frac{6}{25}\right\}\right\}$$

We now plot the first 1000 two-dimensional Halton vectors:

```
ListPlot[Table[halton[n, 2], {n, 1000}], Frame → True]
```



Compare this graph with the plot of 1000 pseudorandom numbers:

```
ListPlot[RandomReal[{0, 1}, 1000], Frame → True]
```



It can be seen from the previous graphs that the Halton vectors are distributed more evenly than the pseudorandom vectors. It is indeed this evenness, or uniformity, that makes the low-discrepancy sequences work better than the pseudorandom numbers in general.

■ Scrambled Van der Corput and Halton Sequences

The uniformity of the Halton sequence, however, degrades quickly as the dimension, hence the prime bases, increases. For example, if we plot the first 1000 Halton vectors in bases 227 and 229, corresponding to the 49th and 50th prime numbers, we get the following:

```
ListPlot[Table[{vanderCorput[n, 227], vanderCorput[n, 229]}, {n, 1000}], Frame → True]
```



This correlation will certainly result in very poor numerical results, if we were using this sequence in simulation, and the simulation of the problem at hand heavily depended on 49th and 50th components.

The uniformity of the van der Corput numbers can be improved by permuting the coefficients in the digit expansion of n in base b. The scrambled van der Corput sequence in base b is defined by $s_b(n) = \sum_{j=0}^{\infty} \sigma\left(a_j(n)\right) b^{-j-1}$, where σ is a permutation on the digits {0,1,...,b-1}. The following permutations, defined recursively, are given in Faure [2]:

```
In[5]:= perm[k_, 2] := perm[k, 2] = k;
        perm[k_, b_ /; EvenQ[b]] := perm[k, b] =
          If[k <= (b / 2) - 1, 2 * perm[k, b / 2], 2 * perm[k - b / 2, b / 2] + 1];
        perm[k_, b_ /; OddQ[b]] := perm[k, b] =
          Module[{c}, c = Floor[b / 2];
           Which[k == c, k,
            (0 <= k < c && 0 <= perm[k, b - 1] < c), perm[k, b - 1],
            (0 <= k < c && c <= perm[k, b - 1] < 2 * c), perm[k, b - 1] + 1,
            (c < k <= 2 * c && 0 <= perm[k - 1, b - 1] < c), perm[k - 1, b - 1],
            (c < k <= 2 * c && c <= perm[k - 1, b - 1] < 2 * c), perm[k - 1, b - 1] + 1] ];
```

Using these permutations, we now construct the scrambled van der Corput sequence in base b.

```
In[8]:= scrvanderCorput[n_Integer, b_Integer] :=
         Map[perm[#, b] &, IntegerDigits[n, b]].(b^Range[-Floor[Log[b, n] + 1], -1])

        Attributes[scrvanderCorput] = Listable;
```

The scrambled Halton sequence is constructed using the scrambled van der Corput numbers.

```
In[10]:= scrhalton[n_Integer, s_Integer] := scrvanderCorput[n, Prime[Range[s]]];

         Table[scrhalton[i, 3], {i, 1, 11}]
```

$$\left\{\left\{\frac{1}{2}, \frac{1}{3}, \frac{3}{5}\right\}, \left\{\frac{1}{4}, \frac{2}{3}, \frac{2}{5}\right\}, \left\{\frac{3}{4}, \frac{1}{9}, \frac{1}{5}\right\}, \left\{\frac{1}{8}, \frac{4}{9}, \frac{4}{5}\right\}, \left\{\frac{5}{8}, \frac{7}{9}, \frac{3}{25}\right\}, \left\{\frac{3}{8}, \frac{2}{9}, \frac{18}{25}\right\}, \left\{\frac{7}{8}, \frac{5}{9}, \frac{13}{25}\right\}, \left\{\frac{1}{16}, \frac{8}{9}, \frac{8}{25}\right\}, \left\{\frac{9}{16}, \frac{1}{27}, \frac{23}{25}\right\}, \left\{\frac{5}{16}, \frac{10}{27}, \frac{2}{25}\right\}, \left\{\frac{13}{16}, \frac{19}{27}, \frac{17}{25}\right\}\right\}$$

These are the first six three-dimensional scrambled Halton vectors:

In[13]:=
```
ListPlot[Table[{scrvanderCorput[n, 227], scrvanderCorput[n, 229]}, {n, 1000}], Frame → True]
```

Out[13]=


■ Kocis-Whiten permutations

Let's say we want to find the permutations for the van der Corput sequence in base 31, which is the 11th prime number.

Compute

```
den = Log[31] / Log[2.]
```

```
4.9542
```

Compute the base 2 van der Corput numbers from 0 to the 31th.

```
l = Table[vanderCorput[n, 2], {n, 0, 31}]
```

$$\left\{0, \frac{1}{2}, \frac{1}{4}, \frac{3}{4}, \frac{1}{8}, \frac{5}{8}, \frac{3}{8}, \frac{7}{8}, \frac{1}{16}, \frac{9}{16}, \frac{5}{16}, \frac{13}{16}, \frac{3}{16}, \frac{11}{16}, \frac{7}{16}, \frac{15}{16}, \frac{1}{32}, \frac{17}{32}, \frac{9}{32}, \frac{25}{32}, \frac{5}{32}, \frac{21}{32}, \frac{13}{32}, \frac{29}{32}, \frac{3}{32}, \frac{19}{32}, \frac{11}{32}, \frac{27}{32}, \frac{7}{32}, \frac{23}{32}, \frac{15}{32}, \frac{31}{32}\right\}$$

Multiply these numbers by 32.

```
per = (2^Ceiling[den]) * l
```

```
{0, 16, 8, 24, 4, 20, 12, 28, 2, 18, 10, 26, 6, 22,
 14, 30, 1, 17, 9, 25, 5, 21, 13, 29, 3, 19, 11, 27, 7, 23, 15, 31}
```

If there is a number greater than or equal to 31 in the above list, skip it:

```
per = Drop[per, Flatten[Position[per, 31]]]
```

```
{0, 16, 8, 24, 4, 20, 12, 28, 2, 18, 10, 26, 6, 22,
 14, 30, 1, 17, 9, 25, 5, 21, 13, 29, 3, 19, 11, 27, 7, 23, 15}
```

```
Length[per]
```

```
31
```

This is the permutation for base 31. The following is a compact code:

```
kw[b_] := kw[b] = Module[{d, per, rl},
   d = Ceiling[Log[b] / Log[2.]];
   per = (2^d) * Table[vanderCorput[n, 2], {n, 0, 2^d - 1}];
   rl = Select[per, # ≥ b &];
   Do[
    per = Delete[per, Flatten[Position[per, rl[[i]]]]], {i, 1, Length[rl]}];
   per]
```

```
kw[31]
```

```
{0, 16, 8, 24, 4, 20, 12, 28, 2, 18, 10, 26, 6, 22,
 14, 30, 1, 17, 9, 25, 5, 21, 13, 29, 3, 19, 11, 27, 7, 23, 15}
```

```
Length[kw[31]]
```

```
31
```

```
kw[10]
```

```
{0, 8, 4, 2, 6, 1, 9, 5, 3, 7}
```

```
kwscrvanderCorput[n_Integer, b_Integer] :=
 Map[kw[b][[#]] &, IntegerDigits[n, b] + 1].(b^Range[-Floor[Log[b, n] + 1], -1])
```

```
Attributes[kwscrvanderCorput] = Listable;
```

```
Table[kwscrvanderCorput[k, 7], {k, 1, 6}]
```

$$\left\{\frac{4}{7}, \frac{2}{7}, \frac{6}{7}, \frac{1}{7}, \frac{5}{7}, \frac{3}{7}\right\}$$

```
Table[scrvanderCorput[k, 7], {k, 1, 6}]
```

$$\left\{\frac{2}{7}, \frac{5}{7}, \frac{3}{7}, \frac{1}{7}, \frac{4}{7}, \frac{6}{7}\right\}$$

The scrambled Halton sequence is constructed using the scrambled van der Corput numbers.

```
kwscrhalton[n_Integer, s_Integer] := kwscrvanderCorput[n, Prime[Range[s]]];
```

These are the first six three-dimensional scrambled Halton vectors:

```
kwscrhalton[3, 7]
```

$$\left\{\frac{3}{4}, \frac{2}{9}, \frac{1}{5}, \frac{6}{7}, \frac{2}{11}, \frac{12}{13}, \frac{4}{17}\right\}$$

```
ListPlot[
 Table[{kwscrvanderCorput[n, 227], kwscrvanderCorput[n, 229]}, {n, 1000}], Frame → True]
```